

ЗАНЯТИЕ 11 (1:20)

Проведение документа по нескольким регистрам

ПРОДОЛЖИТЕЛЬНОСТЬ

Ориентировочная продолжительность занятия – 1 час 20 минут.

Зачем нужно проведение документа по нескольким регистрам	296
Добавление еще одного регистра накопления	297
Проведение приходной накладной по двум регистрам	299
Изменение процедуры проведения	299
Команда перехода к записям регистра	303
Проведение документа «Оказание услуги» по двум регистрам	305
Новый реквизит документа	306
Изменение процедуры проведения	308
Контрольные вопросы	313

Это занятие будет посвящено тому, как один и тот же документ может поставлять информацию в различные регистры конфигурации и для чего может понадобиться такая возможность.

На этом занятии мы создадим еще один регистр накопления нашей конфигурации и изменим процедуру проведения документов так, чтобы они записывали необходимые данные как в один, так и в другой регистр.

Также мы подготовим базу для изучения следующей главы.

Зачем нужно проведение документа по нескольким регистрам

До сих пор мы с вами учитывали только количественное движение материалов в ООО «На все руки мастер». Для этих целей мы создали регистр накопления `ОстаткиМатериалов`.

Однако как вы, наверное, догадываетесь, одного только количественного учета совершенно недостаточно для нужд нашего предприятия. Очевидно, что необходимо также знать, какие денежные средства были затрачены на приобретение тех или иных материалов и каковы материальные запасы ООО «На все руки мастер» в денежном выражении.

После того как мы начали автоматизировать наше предприятие, руководство ООО «На все руки мастер» высказало пожелание, чтобы весь суммовой учет материалов велся бы теперь по средней стоимости.

То есть при закупке материалов они должны учитываться в ценах приобретения, а при расходе — по средней стоимости, которая рассчитывается исходя из общей суммы закупок данного материала и общего количества этого материала, находящегося в ООО «На все руки мастер».

Поскольку подобная информация имеет совершенно другую структуру, нежели количественный учет, для хранения данных об общей стоимости тех или иных материалов мы будем использовать еще один регистр накопления `СтоимостьМатериалов`.

Таким образом, документы `ПриходнаяНакладная` и `ОказаниеУслуги` должны будут создавать движения не только в регистре `ОстаткиМатериалов`, но одновременно и в регистре `СтоимостьМатериалов`, отражая изменения суммового учета.

Добавление еще одного регистра накопления

В режиме «Конфигуратор»

Регистр **СтоимостьМатериалов** совсем не сложен, поэтому мы не будем подробно останавливаться на его создании.

Создадим новый объект конфигурации **Регистр накопления** с именем **СтоимостьМатериалов**.

Расширенное представление списка зададим как **Движения** по регистру **Стоимость материалов**. Этот заголовок будет отображаться в окне списка записей регистра.

На закладке **Подсистемы** отметим, что этот регистр будет отображаться в подсистемах **Бухгалтерия**, **УчетМатериалов** и **ОказаниеУслуг**.

На закладке **Данные** создадим для регистра одно измерение – **Материал** типа **СправочникСсылка.Номенклатура** и один ресурс – **Стоимость** типа **Число** длиной 15 и точностью 2.

После создания регистр **СтоимостьМатериалов** должен выглядеть в дереве конфигурации следующим образом (рис. 11.1).




Рис. 11.1. Регистр накопления «Стоимость материалов»

Теперь отредактируем командный интерфейс, чтобы в разделах Бухгалтерия, Оказание услуг и Учет материалов была доступна команда для просмотра нашего регистра накопления.

В дереве объектов конфигурации выделим ветвь Подсистемы, вызовем ее контекстное меню и выберем пункт Все подсистемы.

В открывшемся окне слева в списке Подсистемы выделим подсистему Бухгалтерия.

Справа в списке Командный интерфейс отразятся все команды выбранной подсистемы.

В группе Панель навигации.Обычное включим видимость у команды Стоимость материалов и мышью перетащим ее в группу Панель навигации.См.также (рис. 11.2).




Рис. 11.2. Настройка командного интерфейса подсистем

Аналогично, выделив подсистемы ОказаниеУслуг и УчетМатериалов, в группе Панель навигации.Обычное включим видимость у команды Стоимость материалов и перенесем ее в группу Панель навигации. См. также.

Теперь мы можем приступить к внесению изменений в процедуры проведения наших документов.

Начнем с самого простого – документа Приходная накладная.

Проведение приходной накладной по двум регистрам

В режиме «Конфигуратор»

Изменение процедуры проведения

Откроем в конфигураторе окно редактирования объекта конфигурации Документ ПриходнаяНакладная и перейдем на закладку Движения.

В списке регистров отметим, что документ будет создавать теперь движения и по регистру СтоимостьМатериалов (рис. 11.3).




Рис. 11.3. Создание движений документа «ПриходнаяНакладная» в регистре «Стоимость материалов»

Нажмем кнопку Конструктор движений. На вопрос системы о замещении процедуры проведения документа на новую, сформированную конструктором, ответим утвердительно. Мы ничего

не потеряем, так как конструктор в обработке проведения сформирует движения уже по двум регистрам, а в прежнюю процедуру проведения мы никаких изменений не вносили.

В открывшемся окне конструктора движений мы увидим, что для регистра ОстаткиМатериалов все поля конструктора уже содержат информацию, которую мы задавали ранее при формировании движений для этого регистра (см. раздел «Движения документа»). Над списком Регистры нажмем кнопку Добавить и добавим еще один регистр СтоимостьМатериалов (рис. 11.4).




Рис. 11.4. Конструктор движений регистров

Тип движения регистра СтоимостьМатериалов по умолчанию будет установлен как Приход (пиктограмма со знаком + слева от имени регистра).

В поле выбора Табличная часть выберем табличную часть нашего документа – Материалы.

Список реквизитов документа, который уже заполнен реквизитами шапки документа, автоматически дополнится реквизитами нашей табличной части.

Теперь нажмем кнопку **Заполнить выражения**. В нижнем окне сформируется соответствие полей (измерений и ресурсов) регистра и выражений для их расчета.

Чтобы задать нужное соответствие для поля регистра **Стоимость**, нужно его выделить и в окне **Реквизиты документа** дважды щелкнуть по строке **ТекСтрокаМатериалы.Сумма** (рис. 11.5).




Рис. 11.5. Конструктор движений регистров

Нажмем **ОК**. После этого откроется модуль объекта документа **Приходная накладная**. Откроем процедуру обработчика события **ОбработкаПроведения**.

Мы увидим, что конструктор сформировал новый текст обработчика, содержащий движения по двум регистрам (листинг 11.1).

Листинг 11.1. Движения документа «ПриходнаяНакладная»

```
Процедура ОбработкаПроведения(Отказ, Режим)
//{{_КОНСТРУКТОР_ДВИЖЕНИЙ_РЕГИСТРОВ
// Данный фрагмент построен конструктором.
// При повторном использовании конструктора внесенные вручную
// изменения будут утеряны!!!

// регистр ОстаткиМатериалов Приход
Движения.ОстаткиМатериалов.Записывать = Истина;
Для Каждого ТекСтрокаМатериалы Из Материалы Цикл
```

```

    Движение = Движения.ОстаткиМатериалов.Добавить();
    Движение.ВидДвижения = ВидДвиженияНакопления.Приход;
    Движение.Период = Дата;
    Движение.Материал = ТекСтрокаМатериалы.Материал;
    Движение.Склад = Склад;
    Движение.Количество = ТекСтрокаМатериалы.Количество;
    КонецЦикла;

    // регистр СтоимостьМатериалов Приход
    Движения.СтоимостьМатериалов.Записывать = Истина;
    Для Каждого ТекСтрокаМатериалы Из Материалы Цикл
        Движение = Движения.СтоимостьМатериалов.Добавить();
        Движение.ВидДвижения = ВидДвиженияНакопления.Приход;
        Движение.Период = Дата;
        Движение.Материал = ТекСтрокаМатериалы.Материал;
        Движение.Стоимость = ТекСтрокаМатериалы.Сумма;
    КонецЦикла;

    //}_КОНСТРУКТОР_ДВИЖЕНИЙ_РЕГИСТРОВ
    КонецПроцедуры

```

Мы видим, что фрагмент, добавленный конструктором в процедуру обработки проведения документа для регистра **СтоимостьМатериалов**, аналогичен фрагменту для формирования движений по регистру **ОстаткиМатериалов**.

Однако сразу бросается в глаза, что совершенно не нужно обходить два одинаковых цикла по элементам табличной части **Материалы**. Можно в одном цикле формировать движения сразу по двум регистрам.

Выполним эту небольшую оптимизацию. И еще удалим комментарии, внесенные конструктором.

В результате процедура **ОбработкаПроведения** будет выглядеть следующим образом (листинг 11.2).

Листинг 11.2. Движения документа «ПриходнаяНакладная»

Процедура **ОбработкаПроведения**(Отказ, Режим)

```

    Движения.ОстаткиМатериалов.Записывать = Истина;
    Движения.СтоимостьМатериалов.Записывать = Истина;

    Для Каждого ТекСтрокаМатериалы Из Материалы Цикл

        // Регистр ОстаткиМатериалов Приход
        Движение = Движения.ОстаткиМатериалов.Добавить();
        Движение.ВидДвижения = ВидДвиженияНакопления.Приход;
        Движение.Период = Дата;

```



```

Движение.Материал = ТекСтрокаМатериалы.Материал;
Движение.Склад = Склад;
Движение.Количество = ТекСтрокаМатериалы.Количество;

// Регистр Стоимость Материалов Приход
Движение = Движения.СтоимостьМатериалов.Добавить();
Движение.ВидДвижения = ВидДвиженияНакопления.Приход;
Движение.Период = Дата;
Движение.Материал = ТекСтрокаМатериалы.Материал;
Движение.Стоимость = ТекСтрокаМатериалы.Сумма;

```

КонецЦикла;

КонецПроцедуры

Команда перехода к записям регистра

В заключение отредактируем командный интерфейс формы документа, чтобы в панели навигации формы иметь возможность переходить к списку записей регистра `СтоимостьМатериалов`, связанному с документом.

Для этого откроем форму документа `ПриходнаяНакладная`. В левом верхнем окне перейдем на закладку `Командный интерфейс`. В разделе `Панель навигации` раскроем группу `Перейти` и увидим команду для открытия регистра накопления `Стоимость материалов`. Установим свойство `Видимость` для этой команды (рис. 11.6).




Рис. 11.6. Настройка командного интерфейса формы документа «ПриходнаяНакладная»

В режиме «1С:Предприятие»

В режиме 1С:Предприятие наша задача будет заключаться в том, чтобы провести еще раз (перепровести) все приходные накладные. Это необходимо для того, чтобы эти документы создали новые записи в регистрах в соответствии с алгоритмом проведения, который мы только что изменили.

Запустим «1С:Предприятие» в режиме отладки. Откроем список документов, выполнив команду Приходные накладные в разделе Учет материалов.

Выделим одновременно, используя клавишу Ctrl, все приходные накладные и перепроведем их, выполнив команду Провести из подменю Еще.

Затем откроем первый документ (рис. 11.7) и, выполнив команды перехода к регистрам Остатки материалов и Стоимость материалов, убедимся, что документ создает желаемые записи как в одном (рис. 11.8), так и в другом регистре накопления (рис. 11.9).

N	Материал	Количество	Цена	Сумма
1	Строчный трансформатор GoldStar (материал)	10,000	270,00	2 700,00
2	Строчный трансформатор Samsung (материал)	10,000	600,00	6 000,00
3	Транзистор Philips 2N2369 (материал)	10,000	3,00	30,00

Рис. 11.7. Приходная накладная № 1

Период	Регистратор	Номер строки	Материал	Склад	Количество
+ 12.10.2022 15:31:29	Приходная накладная 000000001...	1	Строчный трансформатор GoldStar (материал)	Основной	10,000
+ 12.10.2022 15:31:29	Приходная накладная 000000001...	2	Строчный трансформатор Samsung (материал)	Основной	10,000
+ 12.10.2022 15:31:29	Приходная накладная 000000001...	3	Транзистор Philips 2N2369 (материал)	Основной	10,000

Рис. 11.8. Записи регистра «Остатки материалов»

Период	Регистратор	Номер строки	Материал	Стоимость
+ 12.10.2022 15:31:29	Приходная накладная 000000001 ...	1	Строчный трансформатор GoldStar (материал)	2 700,00
+ 12.10.2022 15:31:29	Приходная накладная 000000001 ...	2	Строчный трансформатор Samsung (материал)	6 000,00
+ 12.10.2022 15:31:29	Приходная накладная 000000001 ...	3	Транзистор Philips 2N2369 (материал)	30,00

Рис. 11.9. Записи регистра «Стоимость материалов»

Проведение документа «Оказание услуги» по двум регистрам

В заключение мы внесем изменения в процедуру обработки проведения документа ОказаниеУслуги.

При этом мы будем исходить из пожелания, высказанного руководством ООО «На все руки мастер». Суть его заключается в том, что на первом этапе при списании материалов, израсходованных в процессе оказания услуги, должна быть возможность указывать различную стоимость для одного и того же материала, которая рассчитана руководством исходя из текущих конъюнктурных соображений.

Поскольку в документе ОказаниеУслуги у нас отражена только цена номенклатуры, нам понадобится:

1. Добавить в табличную часть документа еще один реквизит, в котором будет указываться стоимость номенклатуры.
2. После этого изменить процедуру проведения документа ОказаниеУслуги.
3. И в заключение в режиме 1С:Предприятие перепровести все эти документы, чтобы отработал новый измененный нами алгоритм проведения документов Оказание услуги.

В режиме «Конфигуратор»

Новый реквизит документа

Откроем в конфигураторе окно редактирования объекта конфигурации Документ ОказаниеУслуги и перейдем на закладку Данные.

Создадим новый реквизит табличной части документа с именем Стоимость, типом Число, длиной 15 и точностью 2, неотрицательное (рис. 11.10).




Рис. 11.10. Изменение документа «ОказаниеУслуги»

После этого откроем форму ФормаДокумента документа ОказаниеУслуги и добавим в табличную часть ПереченьНоменклатуры поле, отображающее новый реквизит Стоимость.

Для этого в правом верхнем окне редактора форм на закладке Реквизиты раскроем реквизит формы Объект (рис. 11.11).

Мы видим, что он содержит все реквизиты документа ОказаниеУслуги.




Рис. 11.11. Изменение формы документа «ОказаниеУслуги»

Найдем в табличной части реквизит **Стоимость** и с помощью мыши перетащим его в окно элементов формы, расположенное слева в верхней части редактора форм.

Новый элемент расположим в структуре элементов формы после поля **Номенклатура**. Оставим свойства элемента формы, предложенные по умолчанию.

Новый реквизит сразу же отобразится в форме документа, расположенной в левом нижнем окне редактора форм (рис. 11.12).




Рис. 11.12. Изменение формы документа «ОказаниеУслуги»

Изменение процедуры проведения

Теперь создадим движения документа **ОказаниеУслуги**.

В окне редактирования объекта конфигурации **Документ ОказаниеУслуги** перейдем на закладку **Движения**. В списке регистров отметим, что документ будет создавать теперь движения и по регистру **СтоимостьМатериалов**.

Чтобы не потерять изменения, которые мы вносили в процедуру обработки проведения этого документа на предыдущем занятии, на этот раз мы не будем использовать конструктор движений документа, а внесем необходимые дополнения прямо в обработчик события **ОбработкаПроведения документа ОказаниеУслуги**.

Перейдем на закладку **Прочее** и откроем модуль объекта. Для этого нажмем кнопку **Модуль объекта**. Откроем процедуру обработчика события **ОбработкаПроведения**.

В самом конце цикла перед строкой **КонецЕсли** добавим строки кода, создающие движения регистра **СтоимостьМатериалов**, производимые документом **ОказаниеУслуги** (листинг 11.3).

Листинг 11.3. Движения документа «ОказаниеУслуги» (фрагмент)

```
// регистр СтоимостьМатериалов Расход
Движение = Движения.СтоимостьМатериалов.Добавить();
Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
Движение.Период = Дата;
Движение.Материал = ТекСтрокаПереченьНоменклатуры.Номенклатура;
Движение.Стоимость = ТекСтрокаПереченьНоменклатуры.Количество *
ТекСтрокаПереченьНоменклатуры.Стоимость;
```

Перед началом цикла установим свойство **Записывать** набора записей движений по этому регистру в значение **Истина**. Удалим комментарии, внесенные конструктором.

В результате процедура **ОбработкаПроведения** будет выглядеть следующим образом (листинг 11.4).

Листинг 11.4. Движения документа «ОказаниеУслуги»

Процедура **ОбработкаПроведения(Отказ, Режим)**

```
Движения.ОстаткиМатериалов.Записывать = Истина;
Движения.СтоимостьМатериалов.Записывать = Истина;
```

Для Каждого **ТекСтрокаПереченьНоменклатуры** Из **ПереченьНоменклатуры**

Цикл

```
Если ТекСтрокаПереченьНоменклатуры.Номенклатура.ВидНоменклатуры =
Перечисления.ВидыНоменклатуры.Материал Тогда
```

```
// Регистр ОстаткиМатериалов Расход
Движение = Движения.ОстаткиМатериалов.Добавить();
Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
Движение.Период = Дата;
Движение.Материал =
    ТекСтрокаПереченьНоменклатуры.Номенклатура;
Движение.Склад = Склад;
Движение.Количество =
    ТекСтрокаПереченьНоменклатуры.Количество;

// Регистр СтоимостьМатериалов Расход
Движение = Движения.СтоимостьМатериалов.Добавить();
Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
Движение.Период = Дата;
Движение.Материал =
    ТекСтрокаПереченьНоменклатуры.Номенклатура;
Движение.Стоимость = ТекСтрокаПереченьНоменклатуры.Количество
    * ТекСтрокаПереченьНоменклатуры.Стоимость;
```

```
КонецЕсли;
КонецЦикла;
```

```
КонецПроцедуры
```

Обратите внимание, что измерение регистра **Стоимость** вычисляется как произведение стоимости и количества, указанных в табличной части документа.

В заключение отредактируем командный интерфейс формы документа, чтобы в панели навигации формы иметь возможность переходить к списку записей регистра **Стоимость Материалов**, связанному с документом.

Для этого откроем форму документа **ОказаниеУслуги**. В левом верхнем окне перейдем на закладку **Командный интерфейс**. В разделе **Панель навигации** раскроем группу **Перейти** и увидим команду для открытия регистра накопления **Стоимость материалов**. Установим свойство **Видимость** для этой команды.

В режиме «1С:Предприятие»

В режиме **1С:Предприятие** наша задача будет заключаться в том, чтобы провести еще раз (перепровести) наш документ оказания услуги. Это необходимо для того, чтобы этот документ создал новые записи в регистрах в соответствии с алгоритмом проведения, который мы только что изменили.

Запустим «1С:Предприятие» в режиме отладки и откроем список документов, выполнив команду Оказание услуг в разделе Оказание услуг.

Откроем документ Оказание услуги № 1 и укажем в нем стоимость резинового шланга – 100 (рис. 11.13).

Оказание услуги 000000001 от 14.10.2022 20:10:11 *

Основное Остатки материалов Стоимость материалов

Провести и закрыть Записать Провести Печать

Номер: 000000001

Дата: 14.10.2022 20:10:11

Склад: Основной

Клиент: Иванов Михаил Юрьевич

Мастер: Деловой Иван Сергеевич

Добавить

N	Номенклатура	Стоимость	Количество	Цена	Сумма
1	Подключение воды (услуга)		1,000	800,00	800,00
2	Шланг резиновый (материал)	100,00	1,000	150,00	150,00

Рис. 11.13. Документ «Оказание услуги № 1»

Проведем документ Оказание услуги № 1 и посмотрим на движения этого документа по регистру Стоимость материалов.

Для этого нажмем кнопку Провести и выполним команду перехода к регистру Стоимость материалов (рис. 11.14).

Оказание услуги 000000001 от 14.10.2022 20:10:11

Основное Остатки материалов **Стоимость материалов**

Движения по регистру Стоимость материалов

Период: 14.10.2022 20:10:11

Период	Регистратор	Номер строки	Материал	Стоимость
14.10.2022 20:10:11	Оказание услуги 000000001 ...	1	Шланг резиновый (материал)	100,00

Рис. 11.14. Записи регистра «Стоимость материалов»

Теперь создадим и проведем еще два документа Оказание услуги.

Для этого в форме списка документов нажмем кнопку Создать или в панели функций раздела Оказание услуг выполним команду Оказание услуги (рис. 11.15).




Рис. 11.15. Создание новых документов

Эти документы понадобятся нам в дальнейшем, поэтому будьте внимательны и обратите внимание на то, что эти документы созданы другими датами (рис. 11.16, 11.17).




Рис. 11.16. Документ «Оказание услуги № 2»

← → ☆ Оказание услуги (создание) *

Основное Остатки материалов Стоимость материалов

Провести и закрыть Записать Провести Печать

Еще -

Номер:

Дата: 16.10.2022 0:00:00

Склад: Основной

Клиент: Роман

Мастер: Симонов Валерий Михайлович

Добавить

Поиск (Ctrl+F)

Еще -

N	Номенклатура	Стоимость	Количество	Цена	Сумма
1	Подключение электричества (услуга)		1,000	800,00	800,00
2	Шланг резиновый (материал)	100,00	2,000	150,00	300,00
3	Кабель электрический (материал)	20,00	1,000	30,00	30,00
4	Ремонт отечественного телевизора (услуга)		1,000	600,00	600,00
5	Строчный трансформатор GoldStar (материал)	270,00	1,000	400,00	400,00
6	Транзистор Philips 2N2369 (материал)	3,00	2,000	7,00	14,00

Рис. 11.17. Документ «Оказание услуги № 3»

Движения документов Оказание услуги № 2 и № 3 по регистру Стоимость материалов должны выглядеть соответственно следующим образом (рис. 11.18, 11.19).

← → ☆ Оказание услуги 000000002 от 16.10.2022 22:09:13

Основное Остатки материалов Стоимость материалов

Движения по регистру Стоимость материалов

Поиск (Ctrl+F)

Q - Еще -

Период	Регистратор	Номер строки	Материал	Стоимость
16.10.2022 22:09:13	Оказание услуги 000000002 ...	1	Строчный трансформатор Samsung (материал)	600,00

Рис. 11.18. Движения документа «Оказание услуги № 2»

← → ☆ Оказание услуги 000000003 от 16.10.2022 22:15:32

Основное Остатки материалов Стоимость материалов

Движения по регистру Стоимость материалов

Поиск (Ctrl+F)

Q - Еще -

Период	Регистратор	Номер строки	Материал	Стоимость
16.10.2022 22:15:32	Оказание услуги 000000003 ...	1	Шланг резиновый (материал)	200,00
16.10.2022 22:15:32	Оказание услуги 000000003 ...	2	Кабель электрический (материал)	20,00
16.10.2022 22:15:32	Оказание услуги 000000003 ...	3	Строчный трансформатор GoldStar (материал)	270,00
16.10.2022 22:15:32	Оказание услуги 000000003 ...	4	Транзистор Philips 2N2369 (материал)	6,00

Рис. 11.19. Движения документа «Оказание услуги № 3»

Контрольные вопросы

- ☑ *Для чего может понадобиться проведение документа по нескольким регистрам?*
- ☑ *Как создать движения документа по нескольким регистрам в обработчике проведения документа?*
- ☑ *Как создать движения документа без использования конструктора движений?*
- ☑ *Как средствами встроенного языка сформировать и записать движения документа в регистр накопления?*
- ☑ *Как добавить в форму документа новый реквизит?*