

ЗАНЯТИЕ 12 (0:40)

Оборотные регистры накопления

ПРОДОЛЖИТЕЛЬНОСТЬ

Ориентировочная продолжительность занятия – 40 минут.

Зачем нужно создавать еще один регистр	316
Что такое оборотный регистр накопления	316
Добавление оборотного регистра накопления	318
Проведение документа «Оказание услуги» по трем регистрам	320
Контрольные вопросы	325

На этом занятии мы с вами познакомимся с еще одним видом регистра накопления – оборотным регистром накопления.

Вы узнаете о некоторых важных принципах выбора измерений и реквизитов регистров накопления.

Мы с вами создадим оборотный регистр накопления и добавим в один из наших документов движения еще и по этому регистру.

Зачем нужно создавать еще один регистр

Продолжим рассматривать работу нашего документа ОказаниеУслуги.

До сих пор мы создавали в регистрах накопления движения только для строк документа, которые содержат материалы. Услуги, содержащиеся в документе, мы никак не учитывали.

Дело в том, что при учете услуг важны совершенно другие критерии, нежели при учете материалов.

Прежде всего, бессмысленно говорить о том, сколько услуг было и сколько их осталось, важна только сумма и количество услуг, которые были оказаны за определенный промежуток времени.

Кроме этого, интересны следующие моменты:

- какие именно услуги были оказаны (чтобы составить рейтинг услуг);
- какому именно клиенту оказывались услуги (чтобы, например, предоставить ему скидку от объема оплаченных ранее услуг);
- какой мастер предоставлял услуги (чтобы начислить ему заработную плату).

Очевидно, что существующие регистры накопления совершенно не подходят для решения таких задач.

Поэтому мы создадим еще одно хранилище данных, которое будет использоваться в нашей программе, – оборотный регистр накопления Продажи.

Что такое оборотный регистр накопления

Когда мы создавали регистры ОстаткиМатериалов и СтоимостьМатериалов, мы специально не останавливались на видах регистров накопления, которые существуют в системе «1С:Предприятие». Сейчас пришло время сказать об этом несколько слов.

Регистры накопления могут быть *регистрами остатков* и *регистрами оборотов*.

Существующие в нашей учебной конфигурации регистры ОстаткиМатериалов и СтоимостьМатериалов являются регистрами остатков.

Если вы помните, при создании отчета Материалы в конструкторе запроса мы видели, что для таких регистров система создает три виртуальные таблицы: таблицу остатков, таблицу оборотов и совокупную таблицу остатков и оборотов.

Оборотный регистр накопления очень похож на знакомый уже нам регистр остатков, но для него понятие «остаток» не имеет смысла. Оборотный регистр накапливает только обороты, остатки ему безразличны. Поэтому единственной виртуальной таблицей, которую будет создавать система для такого регистра, будет таблица оборотов. В остальном оборотный регистр ничем не отличается от регистра остатков.

Следует сказать об одной особенности конструирования регистров накопления, напрямую связанной с возможностью получения остатков. При создании оборотного регистра накопления нет особой сложности в определении того, какие именно данные должны являться измерениями регистра – мы можем назначить в качестве его измерений любые нужные нам данные.

Совсем иная ситуация в случае регистра накопления, поддерживающего накопление остатков. Для него выбор измерений должен выполняться исходя из того, что движения регистра могут быть осуществлены в две стороны: приход и расход. Таким образом, в качестве измерений нужно выбирать те данные, по которым движения точно будут осуществляться как в одну, так и в другую сторону.

Например, если ведется учет материалов в разрезах номенклатуры и склада, очевидно, что и номенклатура, и склад могут быть измерениями, поскольку как приход, так и расход материалов всегда будут осуществляться с указанием конкретной номенклатуры и конкретного склада. Если же в этой ситуации появляется желание отразить учет материалов еще и в разрезе поставщика, то здесь уже нужно исходить из конкретной схемы учета, принятой на предприятии.

Скорее всего, при поступлении материалов поставщик будет указан, а вот при расходовании материалов с большой долей вероятности поставщик указываться не будет. В большинстве случаев это совершенно лишняя информация.

Значит, поставщика следует добавить не как измерение, а как реквизит регистра накопления.

Если же при расходе материалов поставщик будет указываться наверняка, имеет смысл добавить поставщика в измерения регистра.

Иными словами, по каждому из измерений регистра накопления остатков ресурсы обязательно должны изменяться в обе стороны: приход и расход. Не должно существовать таких измерений, по которым осуществляется только приход или только расход.

Нарушение этого принципа построения регистров накопления будет вести к непроизводительному использованию ресурсов системы и как следствие к замедлению работы и падению производительности.

Для реквизитов же регистра этот принцип не важен. По реквизитам регистра ресурсы могут только приходоваться или только расходываться.

Добавление оборотного регистра накопления

В режиме «Конфигуратор»

Теперь, когда мы знаем практически все о регистрах накопления, откроем конфигуратор и создадим новый объект конфигурации Регистр накопления.

Назовем его Продажи и определим вид регистра – Обороты.

Кроме этого, зададим Расширенное представление списка как Движения по регистру Продажи. Этот заголовок будет отображаться в окне списка записей регистра (рис. 12.1).

На закладке Подсистемы отметим, что этот регистр будет отображаться в подсистемах Бухгалтерия, УчетМатериалов и ОказаниеУслуг.

На закладке Данные создадим измерения регистра:

- Номенклатура, тип СправочникСсылка.Номенклатура;
- Клиент, тип СправочникСсылка.Клиенты;
- Мастер, тип СправочникСсылка.Сотрудники.

У регистра будет три ресурса:

- Количество, тип Число, длина 15, точность 3;
- Выручка, тип Число, длина 15, точность 2;
- Стоимость, тип Число, длина 15, точность 2.




Рис. 12.1. Создание оборотного регистра накопления

После создания регистр Продажи должен выглядеть в дереве конфигурации следующим образом (рис. 12.2).




Рис. 12.2. Оборотный регистр накопления «Продажи»

Теперь отредактируем командный интерфейс, чтобы в подсистемах Бухгалтерия, Оказание услуг и Учет материалов была доступна команда для просмотра нашего оборотного регистра накопления.

В дереве объектов конфигурации выделим ветвь Подсистемы, вызовем ее контекстное меню и выберем пункт Все подсистемы.

В открывшемся окне слева в списке Подсистемы выделим подсистему Бухгалтерия.

Справа в списке Командный интерфейс отразятся все команды выбранной подсистемы.

В группе Панель навигации.Обычное включим видимость у команды Продажи и мышью перетащим ее в группу Панель навигации.См. также.

Аналогично, выделив подсистемы ОказаниеУслуг и УчетМатериалов, в группе Панель навигации.Обычное включим видимость у команды Продажи и перенесем ее в группу Панель навигации.См. также.

Проведение документа «Оказание услуги» по трем регистрам

В этом разделе мы сначала изменим процедуру проведения документа ОказаниеУслуги, а затем в режиме 1С:Предприятие перепроведем все эти документы, чтобы отработал новый, измененный нами алгоритм проведения документов Оказание услуги.

В режиме «Конфигуратор»

Откроем окно редактирования объекта конфигурации Документ ОказаниеУслуги и на закладке Движения укажем, что этот документ будет создавать движения еще и по регистру Продажи.

Перейдем на закладку Прочее и откроем модуль документа. Для этого нажмем кнопку Модуль объекта.

Откроем процедуру обработчика события ОбработкаПроведения.

В конце цикла после строки КонецЕсли и перед строкой КонецЦикла добавим строки кода, создающие движения регистра Продажи, производимые документом ОказаниеУслуги (листинг 12.1).

Листинг 12.1. Движения документа «ОказаниеУслуги» (фрагмент)

```
// Регистр Продажи
Движение = Движения.Продажи.Добавить();
Движение.Период = Дата;
Движение.Номенклатура = ТекСтрокаПереченьНоменклатуры.Номенклатура;
Движение.Клиент = Клиент;
Движение.Мастер = Мастер;
Движение.Количество = ТекСтрокаПереченьНоменклатуры.Количество;
Движение.Выручка = ТекСтрокаПереченьНоменклатуры.Сумма;
Движение.Стоимость = ТекСтрокаПереченьНоменклатуры.Стоимость *
ТекСтрокаПереченьНоменклатуры.Количество;
```

Перед началом цикла установим свойство Записывать набора записей движений по этому регистру в значение Истина.

В результате процедура ОбработкаПроведения будет выглядеть следующим образом (листинг 12.2).

Листинг 12.2. Движения документа «ОказаниеУслуги»

Процедура ОбработкаПроведения(Отказ, Режим)

```
Движения.ОстаткиМатериалов.Записывать = Истина;
Движения.СтоимостьМатериалов.Записывать = Истина;
Движения.Продажи.Записывать = Истина;
```

Для Каждого ТекСтрокаПереченьНоменклатуры Из ПереченьНоменклатуры

Цикл

```
Если ТекСтрокаПереченьНоменклатуры.Номенклатура.ВидНоменклатуры =
Перечисления.ВидыНоменклатуры.Материал Тогда
```

```
// Регистр ОстаткиМатериалов Расход
```

```
Движение = Движения.ОстаткиМатериалов.Добавить();
```

```
Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
```

```
Движение.Период = Дата;
```

```
Движение.Материал =
```

```
ТекСтрокаПереченьНоменклатуры.Номенклатура;
```

```
Движение.Склад = Склад;
```

```
Движение.Количество =
```

```
ТекСтрокаПереченьНоменклатуры.Количество;
```

```
// Регистр СтоимостьМатериалов Расход
```

```
Движение = Движения.СтоимостьМатериалов.Добавить();
```

```
Движение.ВидДвижения = ВидДвиженияНакопления.Расход;
```

```
Движение.Период = Дата;
```

```
Движение.Материал =
```

```
ТекСтрокаПереченьНоменклатуры.Номенклатура;
```

```
Движение.Стоимость =
```

```
ТекСтрокаПереченьНоменклатуры.Количество *
```

```
ТекСтрокаПереченьНоменклатуры.Стоимость;
```

```
КонецЕсли;
```

```
// Регистр Продажи
Движение = Движения.Продажи.Добавить();
Движение.Период = Дата;
Движение.Номенклатура =
    ТекСтрокаПереченьНоменклатуры.Номенклатура;
Движение.Клиент = Клиент;
Движение.Мастер = Мастер;
Движение.Количество = ТекСтрокаПереченьНоменклатуры.Количество;
Движение.Выручка = ТекСтрокаПереченьНоменклатуры.Сумма;
Движение.Стоимость =
    ТекСтрокаПереченьНоменклатуры.Стоимость *
    ТекСтрокаПереченьНоменклатуры.Количество;

КонецЦикла;
КонецПроцедуры
```

Все добавленные конструкции вам уже хорошо известны.

Обратите внимание лишь на то, что у оборотного регистра отсутствует свойство ВидДвижения, поскольку отражение вида движения (приход или расход) имеет смысл лишь при учете остатков. В случае регистра оборотов нас интересует только значение, которое должно быть записано в ресурс регистра.

Также заметьте, что мы разместили команды, создающие движения в регистре Продажи, в конце цикла обхода строк табличной части документа, после условия выполнения цикла только для материалов. Это важно, так как движения в этом регистре создаются как для материалов, так и для услуг.

В заключение отредактируем командный интерфейс формы документа, чтобы в панели навигации формы иметь возможность переходить к списку записей регистра Продажи, связанному с документом.

Для этого откроем форму документа ОказаниеУслуги. В левом верхнем окне перейдем на закладку Командный интерфейс. В разделе Панель навигации раскроем группу Перейти и увидим команду для открытия регистра накопления Продажи.

Установим свойство Видимость для этой команды.

В режиме «1С:Предприятие»

В режиме 1С:Предприятие нам нужно перепровести все документы оказания услуг и проверить, что они создают правильные движения в регистре Продажи.

Запустим «1С:Предприятие» в режиме отладки и откроем по очереди каждый документ Оказание услуги.

Нажмем Провести и перейдем к списку движений этих документов по регистру Продажи. Они должны иметь следующий вид (рис. 12.3, 12.4, 12.5, 12.6, 12.7, 12.8).

← → ☆ Оказание услуги 000000001 от 14.10.2022 20:10:11

Основное Остатки материалов **Продажи** Стоимость материалов

Движения по регистру Продажи

Поиск (Ctrl+F) x Q - Еще -

Период	Регистратор	Номер строки	Номенклатура	Клиент	Мастер
14.10.2022 20:10:11	Оказание услуги 000000001 ...	1	Подключение воды (услуга)	Иванов Михаил Юрьевич	Деловой Иван Сергеевич
14.10.2022 20:10:11	Оказание услуги 000000001 ...	2	Шланг резиновый (материал)	Иванов Михаил Юрьевич	Деловой Иван Сергеевич

Рис. 12.3. Движения документа «Оказание услуги № 1» в регистре «Продажи»

← → ☆ Оказание услуги 000000001 от 14.10.2022 20:10:11 x

Основное Остатки материалов **Продажи** Стоимость материалов

Движения по регистру Продажи

Поиск (Ctrl+F) x Q - Еще -

Клиент	Мастер	Количество	Выручка	Стоимость
Иванов Михаил Юрьевич	Деловой Иван Сергеевич	1,000	800,00	
Иванов Михаил Юрьевич	Деловой Иван Сергеевич	1,000	150,00	100,00

Рис. 12.4. Движения документа «Оказание услуги № 1» в регистре «Продажи»

← → ☆ Оказание услуги 000000002 от 16.10.2022 22:09:13

Основное Остатки материалов **Продажи** Стоимость материалов

Движения по регистру Продажи

Поиск (Ctrl+F) x Q - Еще -

Период	Регистратор	Номер строки	Номенклатура	Клиент	Мастер
16.10.2022 22:09:13	Оказание услуги 000000002 ...	1	Ремонт импортного телевизора (услуга)	Спиридонова Галина	Гусаков ...
16.10.2022 22:09:13	Оказание услуги 000000002 ...	2	Строчный трансформатор Samsung (материал)	Спиридонова Галина	Гусаков ...

Рис. 12.5. Движения документа «Оказание услуги № 2» в регистре «Продажи»

← → ☆ Оказание услуги 000000002 от 16.10.2022 22:09:13

Основное Остатки материалов **Продажи** Стоимость материалов

Движения по регистру Продажи

Поиск (Ctrl+F) × 🔍 Ещё ▾

Клиент	Мастер	Количество	Выручка	Стоимость
Спиридонова Галина	Гусаков Николай Дмитриевич	1,000	800,00	
Спиридонова Галина	Гусаков Николай Дмитриевич	1,000	900,00	600,00

Рис. 12.6. Движения документа «Оказание услуги № 2» в регистре «Продажи»

← → ☆ Оказание услуги 000000003 от 16.10.2022 22:15:32

Основное Остатки материалов **Продажи** Стоимость материалов

Движения по регистру Продажи

Поиск (Ctrl+F) × 🔍 Ещё ▾

Период	Регистратор	Номер строки	Номенклатура	Клиент	Мастер
16.10.2022 22:15:32	Оказание услуги 000000003 ...	1	Подключение электричества (услуга)	Роман	Симонов Валерий ...
16.10.2022 22:15:32	Оказание услуги 000000003 ...	2	Шланг резиновый (материал)	Роман	Симонов Валерий ...
16.10.2022 22:15:32	Оказание услуги 000000003 ...	3	Кабель электрический (материал)	Роман	Симонов Валерий ...
16.10.2022 22:15:32	Оказание услуги 000000003 ...	4	Ремонт отечественного телевизора (услуга)	Роман	Симонов Валерий ...
16.10.2022 22:15:32	Оказание услуги 000000003 ...	5	Строчный трансформатор GoldStar (материал)	Роман	Симонов Валерий ...
16.10.2022 22:15:32	Оказание услуги 000000003 ...	6	Транзистор Philips 2N2369 (материал)	Роман	Симонов Валерий ...

Рис. 12.7. Движения документа «Оказание услуги № 3» в регистре «Продажи»

← → ☆ Оказание услуги 000000003 от 16.10.2022 22:15:32 ×

Основное Остатки материалов **Продажи** Стоимость материалов

Движения по регистру Продажи

Поиск (Ctrl+F) × 🔍 Ещё ▾

Клиент	Мастер	Количество	Выручка	Стоимость
Роман	Симонов Валерий Михайлович	1,000	800,00	
Роман	Симонов Валерий Михайлович	2,000	300,00	200,00
Роман	Симонов Валерий Михайлович	1,000	30,00	20,00
Роман	Симонов Валерий Михайлович	1,000	600,00	
Роман	Симонов Валерий Михайлович	1,000	400,00	270,00
Роман	Симонов Валерий Михайлович	2,000	14,00	6,00

Рис. 12.8. Движения документа «Оказание услуги № 3» в регистре «Продажи»

На рисунках видно, что в регистр Продажи попадают записи о любой номенклатуре, которая является как материалом, так и услугой.

Теперь у нас есть практически вся необходимая информация для анализа деятельности ООО «На все руки мастер».

На следующем занятии мы займемся с вами тем, что создадим несколько отчетов, представляющих нам итоговую информацию о работе предприятия.

Контрольные вопросы

- ☒ *Что такое оборотный регистр накопления?*
- ☒ *В чем отличие между регистром накопления остатков и оборотным регистром накопления?*
- ☒ *Как выбирать реквизиты и измерения при создании регистров накопления?*
- ☒ *Как создать оборотный регистр накопления?*